

An evaluation of diffusion bonding of zircaloy components is performed. The main objective is to determine which of three elements – nickel, iron, or copper – is the best bonding agent. A series of zircaloy components are bonded using each of the possible bonding agents. Due to significant variation in components machined from different ingots, a randomized block design is used, blocking on the ingots. Two components from each ingot are bonded together using each of the three agents, and the pressure (in units of 1000 pounds per square inch) required to separate the bonded components is measured.

```
data rcb;
```

```
  input ingot metal $ pres @@;
```

```
  datalines;
```

```
1 n 67.0 1 i 71.9 1 c 72.2 2 n 67.5 2 i 68.8 2 c 66.4 3 n 76.0 3 i 82.6 3 c 74.5 4 n 72.7 4 i 78.1 4 c 67.3
```

```
5 n 73.1 5 i 74.2 5 c 73.2 6 n 65.8 6 i 70.8 6 c 68.7 7 n 75.6 7 i 84.9 7 c 69.0
```

```
;
```

```
proc mixed data=rcb;
```

```
  class ingot metal;
```

```
  model pres=metal / solution cl;
```

```
  random ingot;
```

```
  lsmeans metal / pdiff adj=tukey;
```

```
  estimate 'nickel mean' intercept 1 metal 0 0 1;
```

```
  estimate 'copper vs iron' metal 1 -1 0;
```

```
  contrast 'copper vs iron' metal 1 -1 0;
```

```
run;
```

```
proc glm data=rcb;
```

```
  class ingot metal;
```

```
  model pres=ingot metal;
```

```
  lsmeans metal/stderr pdiff;
```

```
  estimate 'nickel mean' intercept 1 metal 0 0 1;
```

```
  estimate 'copper vs iron' metal 1 -1 0;
```

```
  contrast 'copper vs iron' metal 1 -1 0;
```

```
  random ingot;
```

```
run;
```

```
proc varcomp data=rcb;
```

```
  class ingot metal;
```

```
  model pres=metal ingot / fixed=1;
```

```
run;
```

The Mixed Procedure

Class	Levels	Values
ingot	7	1 2 3 4 5 6 7
metal	3	c i n

Covariance Parameter Estimates

Cov Parm	Estimate
ingot	11.4478
Residual	10.3716

Solution for Fixed Effects

Effect	metal	Estimate	Standard Error	DF	t Value	Pr > t	Alpha
Intercept		71.1000	1.7655	6	40.27	<.0001	0.05
metal	c	-0.9143	1.7214	12	-0.53	0.6050	0.05
metal	i	4.8000	1.7214	12	2.79	0.0164	0.05
metal	n	0	.	.	.	.	.

Solution for Fixed Effects

Effect	metal	Lower	Upper
Intercept		66.7799	75.4201
metal	c	-4.6650	2.8364
metal	i	1.0493	8.5507
metal	n	.	.

Type 3 Tests of Fixed Effects

Effect	Num DF	Den DF	F Value	Pr > F
metal	2	12	6.36	0.0131

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > t
nickel mean	71.1000	1.7655	12	40.27	<.0001
copper vs iron	-5.7143	1.7214	12	-3.32	0.0061

Contrasts

Label	Num DF	Den DF	F Value	Pr > F
copper vs iron	1	12	11.02	0.0061

Least Squares Means

Effect	metal	Estimate	Standard Error	DF	t Value	Pr > t
metal	c	70.1857	1.7655	12	39.75	<.0001
metal	i	75.9000	1.7655	12	42.99	<.0001
metal	n	71.1000	1.7655	12	40.27	<.0001

Differences of Least Squares Means

Effect	metal	_metal	Estimate	Standard Error	DF	t Value	Pr > t	Adjustment
metal	c	i	-5.7143	1.7214	12	-3.32	0.0061	Tukey-Kramer
metal	c	n	-0.9143	1.7214	12	-0.53	0.6050	Tukey-Kramer
metal	i	n	4.8000	1.7214	12	2.79	0.0164	Tukey-Kramer

Differences of Least Squares Means

Effect	metal	_metal	Adj P
metal	c	i	0.0156
metal	c	n	0.8578
metal	i	n	0.0404

The GLM Procedure

Class Level Information

Class	Levels	Values
ingot	7	1 2 3 4 5 6 7
metal	3	c i n

Number of Observations Read 21
 Number of Observations Used 21
 The SAS System

The GLM Procedure

Dependent Variable: pres

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	8	400.1904762	50.0238095	4.82	0.0076
Error	12	124.4590476	10.3715873		
Corrected Total	20	524.6495238			

R-Square	Coeff Var	Root MSE	pres Mean
0.762777	4.448490	3.220495	72.39524

Source	DF	Type I SS	Mean Square	F Value	Pr > F
ingot	6	268.2895238	44.7149206	4.31	0.0151
metal	2	131.9009524	65.9504762	6.36	0.0131

Source	DF	Type III SS	Mean Square	F Value	Pr > F
ingot	6	268.2895238	44.7149206	4.31	0.0151
metal	2	131.9009524	65.9504762	6.36	0.0131

Least Squares Means

metal	pres LSMEAN	Standard Error	Pr > t	LSMEAN Number
c	70.1857143	1.2172327	<.0001	1
i	75.9000000	1.2172327	<.0001	2
n	71.1000000	1.2172327	<.0001	3

Least Squares Means for effect metal
Pr > |t| for H0: LSMean(i)=LSMean(j)

Dependent Variable: pres

i/j	1	2	3
1		0.0061	0.6050
2	0.0061		0.0164
3	0.6050	0.0164	

NOTE: To ensure overall protection level, only probabilities associated with pre-planned comparisons should be used.

Source	Type III Expected Mean Square
ingot	Var(Error) + 3 Var(ingot)
metal	Var(Error) + Q(metal)

Contrast	Contrast Expected Mean Square
copper vs iron	Var(Error) + Q(metal)

Dependent Variable: pres

Contrast	DF	Contrast SS	Mean Square	F Value	Pr > F
copper vs iron	1	114.2857143	114.2857143	11.02	0.0061

Parameter	Estimate	Standard Error	t Value	Pr > t
nickel mean	71.1000000	1.21723265	58.41	<.0001
copper vs iron	-5.7142857	1.72142692	-3.32	0.0061

Variance Components Estimation Procedure

Class Level Information

Class	Levels	Values
ingot	7	1 2 3 4 5 6 7
metal	3	c i n

Number of Observations Read	21
Number of Observations Used	21

MIVQUE(0) Estimates

Variance Component	pres
Var(ingot)	11.44778
Var(Error)	10.37159

*Summary Comparison of Syntax and Output for PROC GLM, PROC MIXED,
and PROC VARCOMP*

Statement	PROC MIXED	PROC GLM	PROC VARCOMP
CLASS	list classification variables	same	same
MODEL	specify dependent variable and list fixed effect	specify dependent variable and list all terms in model	specify dependent variable and list all terms in model, fixed effects preceding random
RANDOM	specify random effects	obtain table of expected mean squares	not applicable
LSMEANS	estimate means for fixed effect factors	estimate means for fixed effect factors	not applicable
ESTIMATE	estimate linear combination of model terms	estimate linear combination of model terms	not applicable
CONTRAST	test set of linear combinations of model terms	test set of linear combination of model terms	not applicable